Introduction to GAMS:
Summation Notation with GAMS

Dhazn Gillig
&
Bruce A. McCarl

Department of Agricultural Economics
Texas A&M University

Using Summation Notation With GAMS

Materials are drawn from Appendix A in McCarl and Spreen's text book on

http://ageco.tamu.edu/faculty/mccarl/mccspr/newa1.pdf

Using Summation Notation With GAMS – Summation Mechanics

GAMS summation notation involves subscripted items. For illustrative purposes, let's define some data

$x_1 = 1$	$y_{11} = 2$	$y_{12} = 3$
$x_2 = 2$	$y_{21} = 4$	$y_{22} = 1$
$x_3 = 3$	$y_{31} = 1$	$y_{32} = 4.$

Now let us define a variety of summation expressions.

(1). Sum of an Item: Suppose we wish sum up all values of x. This would be written as

$$\sum_{i} x_i \qquad \sum_{i=1}^{3} x_i = x_1 + x_2 + x_3 \qquad = 1 + 2 + 3 = 6$$

or in **GAMS** SumThis1 = SUM(I , X(I)) ;

This says sum up x(I) for all values of I and in summation notation is

Using Summation Notation With GAMS – Summation Mechanics

(2). Multiple Sums: Sums over two indices consider all combinations of those items $\sum_i \sum_j y_{ij} = y_{11} + y_{12} + y_{21} + y_{22} + y_{31} + y_{32} = 15.$

The equivalent GAMS expression is

SumThis2 = SUM((I , J), Y(I,J)) ;

(3). Sum of Two Items: Suppose we wished to sum over two items completely where they share a subscript

$$\sum_{i=1}^{3} (x_i + \sum_{j=1}^{2} y_{ij}) = \sum_i (x_i + \sum_j y_{ij}) = \sum_i x_i + \sum_i \sum_j y_{ij}$$

$$= x_1 + y_{11} + y_{12} + x_2 + y_{21} + y_{22} + x_3 + y_{31} + y_{32} = 21$$

The equivalent GAMS expression is as follows

SumThis3 = SUM(I, X(I) + SUM(J, Y(I, J))) ;

Or SumThis3 = SUM(I, X(I)) + SUM((I,J), Y(I,J)) ;

Using Summation Notation With GAMS – Summation Mechanics

On the other hand, if we wished to sum slices of the data into an item that is itself defined over an index like adding up all of the j cases associated with each index I then

$$A_i = x_i + \sum_j y_{ij} = x_i + y_{i1} + y_{i2}$$

Or in GAMS A(I) = X(I) + SUM(J, Y(I,J)) ;

Sums over common subscripts can be collapsed or taken apart

$$\sum_i (x_i + z_i) = \sum_i x_i + \sum_i z_i$$

Or in GAMS SumThis4 = SUM(I, X(I) +Z(I)) ;

or SumThis4 = SUM(I, X(I)) + SUM(I, Z(I)) ;

Using Summation Notation With GAMS – Rules

Proper equations must obey certain rules. The rules depend on whether the final result is an unsubscripted scalar or a subscripted family of results.

(1). For a Scalar Equation

All subscripts must be dealt with in each term. A proper equation follows

$$B = \sum_i \sum_j \sum_k p_{ijk} + \sum_m \sum_n q_{mn} \, .$$

However, the following equations are wrong

$$C = p_{ijk} + q_{mn}$$ subscripts i,j,k,m,n are not dealt with

$$D = \sum_j \sum_i p_{ijk} + \sum_m \sum_n q_{mn}$$ k subscript is not dealt with in first term.

Or, in GAMS

```
EQB1..     B =E= SUM((I,J,K),P(I,J,K)) + SUM((M,N), Q(M,N));
EQB2..     c =E= P(I,J,K) + Q(M,N);
EQB3..     d =E= SUM((I,J), P(I,J,K)) + SUM((M,N), Q(M,N));
```

The last two equation expressions are incorrect and GAMS will show error message $149 "UNCONTROLLED SET ENTERED AS CONSTANT"

(2). For a Family of Equations

Rule 1. Family defining subscripts must be specified. This indicates the set of conditions over which the equations exist -- a "for all" condition.

Suppose an equation sets C's equal to 2 for all members of set I.

$C_i = 2$ for all i or $C_i = 2$ for i = 1,2, ... n.

In GAMS c(i) = 2;

Similarly, if we wish to set a 2 dimensional variable

Algebra		GAMS
$D_{ij} = 2$	for all i and j,	D(i,j) = 2;
$E1_{ij} = F_i$	for all i and j.	E1(i,j) = F(i);
$E2_{ij} = E1_i$	for all i and j.	E2(i,j) = E1(i);

GAMS automatically handles all subscript cases

The same holds in an equation specification

```
EQUATIONS
      EQC(I)        EQUATION C
      EQD(I,J)      EQUATION D
      EQE1(I,J)     EQUATION E1 ;
EQC(I)..            C(I)     =E= 2;
EQD(I,J)..          D(I,J)   =E= 2;
EQE1(I,J)..         E1(I,J) =E= F(I);
```

Rule 2. **When writing an equation defined over sets (with a for all statement), all subscripts which are not in the equation "for all" definition must be dealt with (summed over) in the terms of the equation.**

Consequently, it is proper to write

$\sum_k p_{ijk} = H_i$ for all i and j **eq1(I,j).. SUM(k, p(I,j,k)) =E= h(I) ;**

$\sum_j \sum_k p_{ijk} = G_i$ for all i **eq2(I).. SUM((j,k),p(I,j,k)) =E= g(I) ;**

$p_{ijk} = G_i$ for all i, j, k **eq3(I,j,k).. p(I,j,k) =E= g(I) ;**

But is improper to write

$\sum_k p_{ijk} = H_i$ for all i **eq4(I).. SUM(k,p(I,j,k)) =E= g(I);**

In such a case GAMS will give error message $149 saying "UNCONTROLLED SET ENTERED AS CONSTANT" . Why?

Rule 3. In any term in an equation, the result after executing the mathematical operations must be of a dimension less than or equal to the family definition in the **for all** statement.

For example, it is proper to write

$$\sum_{j}\sum_{k} r_{ijkm} + \sum_{j} s_{ij} = N_{im} \text{ for all } i, m$$

Or EQN(I,M).. SUM((J,K), R(I,J,K,M))+ SUM (J,S(I,J)) =E= N(I,M) ;

What is different about this term?

but wrong to write

$p_{ijk} = L2$ for all i or EQ22(i).. P(I,j,k) =E= L2;

What is wrong in this term?

Again you would get the error message $149 "UNCONTROLLED SET ENTERED AS CONSTANT."

Note when the dimension is less than the family definition this implies the same term appears in multiple equations. For example, in the equation

$$2 + \sum_j \sum_k p_{ijk} + \sum_j s_{ijm} = O_{im} \qquad \text{for all i and m,}$$

Or in GAMS

EQUATION EQO(I,M) EQUATION O ;

EQO(I,M).. 2 + SUM((J,K), P(I,J,K)) + SUM(J, S(I,J,M)) =E= O(I,M);

the term involving p differs in all the equations associated with the set element i appears but given an i is the same for every case of m. The 2 is the same for all i and m

Is this OK?

Rule 4. In an equation you can never sum over the parameter that determines the family of equations or the for all condition. It is certainly wrong to write

$$\sum_i W_i = N_i \text{ for all } i$$

```
EQW(I)..   Sum(I,W(I))   =E=   N(I) ;
```

$$W_i = \sum_i N_i \text{ for all } i$$

```
W(I)= Sum(I, N(I)) ;
```

Such statements will result in error message $125 which says "SET IS UNDER CONTROL ALREADY." Why?